# Technical Supplement

# Methodology for Modelling the Revenue Impacts of Rollover Relief

**Prepared for:**

**Securities Institute of Australia**

**Prepared by:**

**Access Economics**

**January 1999**

# Technical Supplement
# Methodology for modelling the revenue impacts of rollover relief

Previous sections have discussed key parameters and recommended settings for these parameters. This Technical Supplement discusses how these assumptions are combined to produce a bottom line.

Access Economics has constructed two models for the purpose of estimating the revenue effect of extending rollover relief to share-swap mergers and demergers. The two models have many similar features, hence much of the discussion below relates to both share-swap mergers and demergers. The two models diverge mostly in the settings of the various parameters (discussed in an earlier section), rather than the actual calculations. Where there are differences in the calculations, it is noted in the text.

The modelling is contained in the Excel 97 file **RolloverReliefModel.xls**. There are several worksheets and charts within this spreadsheet file. Sheets and charts that begin with 'Mer' relate to share-swap mergers. Sheets and charts that begin with 'Dem' relate to demergers. The remaining sheets and charts are common to both. There are both colour and black & white versions of the charts, for ease of printing.

The calculations are performed on a gross-flows basis. That is, for all the shares in all the companies within the scope, the model calculates the total capital gain revenue under the current policies and the total capital gain revenue under the rollover relief policy to derive the impact on revenue. Modelling the gross flows ensure subtleties such as compounding and behavioural change are better accounted for.

The modelling work can be roughly divided into two parts:

1. Developing generic functions to calculate capital gains realisation ratios under various circumstances. This involves calculating shares prices, inflation adjustments and so forth under various assumptions and policies. These calculations take into account the profile of shareholders' acquisitions and disposals of shares pre- and post-merger, as follows:

   - For the no-policy change base case:

     - the capital gains realised (as a proportion of market capitalisation) when shares are involuntarily disposed of in a share-swap merger or demerger (calculated by the CG_Disp(...) function discussed on page 7); and
     - the capital gains realised (as a proportion of market capitalisation) when shares are voluntarily disposed of in the years following a share-swap merger or demerger (calculated by the CG_Acqu(...) function discussed on page 11).
     - the capital gains realised (as a proportion of market capitalisation) on shares in companies that do not merge under current policies (but would merge under rollover relief). This uses the CG_Roll(...) function (as there is no involuntary disposal of shares) but in this case the merger/demerger gain is zero as no merger/demerger occurred.

   - For the rollover relief case, the capital gains realised (as a proportion of market capitalisation) when shares are voluntarily disposed of in the years following a share-swap merger or demerger, where CGT rollover relief is afforded (this is the CG_Roll(...) function discussed on page 13).

2. Grossing up the realisation ratios from Part 1 using estimates of the pre-merger market capitalisation of target companies in the case of share-swap mergers, or using estimates of the pre-demerger market capitalisation of companies in the case of demergers. An allowance is made for merger/demerger gains resulting from the merger/demerger activity. Part 2 also calculates the amount of tax revenue from capital gains realisations, allowing for the various tax rates paid by the mix of different types of

investors, offsetting capital losses, changes in activity resulting from the policy change, and so on. Revenue on capital gains that is picked up through the income tax net is also accounted for. Revenue from higher profits (as implied by the merger/demerger gain) is also accounted for.

## *Part 1 of the model – capital gains realisation ratios*

This part of the model develops functions for evaluating capital gains realisation ratios for share-swap mergers and demergers under alternative tax treatments.

The model is constructed in Excel 97. The calculations for part 1 are implemented using custom functions written in Visual Basic. To understand how the model calculates capital gains realisation ratios it is necessary to have an understanding of how to write custom functions in Excel.

### A crash course in Excel custom functions

People familiar with Excel spreadsheets will no doubt be familiar with many of Excel's built-in functions. These include SUM(...), NPV(...), and so on. However, there are often functions we might like to use that are not built-in to Excel. The Visual Basic programming facility in Excel allows users to create their own functions. Once these functions have been created, they operate just like built-in Excel functions.

For example, suppose a modeller needs a function that raises a series of numbers to a given power then sums the series and returns the result. Excel does not have this function built-in so it will need to be custom-made:

- Open Excel 97 and ensure there is a blank workbook open (probably called Book1), then open the Visual Basic editor (hit Alt+F11, or use the menu Tools, Macro, Visual Basic Editor).

- Book1 should appear in a pane down the left-hand side of the editor. Click on Book1 in this left-hand pane and select the menu Insert, Module. A new window will appear, probably titled Book1 – Module1 (Code). Note: if the words `Option Explicit` happen to appear at the top of the new window, delete them so the new window is completely empty.

- In this new window, type the following code (Visual Basic will automatically format and insert some words):

```
Function PowerSUM(TheData, ThePower)
  For i = 1 To TheData.Count
    PowerSUM = PowerSUM + TheData(i) ^ ThePower
  Next i
End Function
```

- This function will go through a range of cells, (which are referred to as `TheData` in the Visual Basic code), raise the value in each cell to `ThePower` and accumulate the result. The Visual Basic editor should look like the following graphic:




- To try out the custom function, switch back to Excel (hit Atl+F11, or use the menu View, Microsoft Excel).

- On a spreadsheet in Book1 type the numbers (say) 1,2,3,4,5,6 in cells A1:A6 respectively. In cell B1 type the formula: =PowerSUM(A1:A6,0.5) (this should return the sum of the square roots)

- In cell B2 type the formula: =PowerSUM(A1:A6,-1) (this should return the sum of the inverses)

- The spreadsheet should look something like the following graphic. The file **PowerSUM.xls** accompanying the model software contains this demonstration if users wish to examine the file on their computer.




There are several advantages to using custom functions:

- It avoids repetition by only having to write a complex capital gains calculation once. Then the modeller can invoke the custom function each time a capital gains realisation calculation is required – this is much safer than the endless copying/filling/repeating of complex formulae often performed in spreadsheet modelling.
- It is more efficient, the model ends up being a few hundred KB instead of probably several MB if the formulas were done the long way on a spreadsheet.
- It is easier to check for errors.
- It is faster to recalculate after a change in assumptions.
- It encourages good programming and modelling habits.

## Calculating capital gains realisation ratios

Before we leap into the functions that calculate capital gains realisation ratios, there are a few preliminary functions that help calculate probabilities given the type of probability distribution selected by the user.

If you wish to see any of the source code for these functions on your computer screen, just hit Alt+F11, or use the menu Tools, Macro, Visual Basic Editor while the model is open in Excel 97. Access Economics has elected not to read- or write-protect this source code so that others may examine our methodology. However, this source code is subject to copyright and should therefore not be copied in whole or in part, unless appropriate arrangements are first discussed with Access Economics.

## The ProbDist(...) function

The ProbDist(...) custom function is used to calculate the proportion of investors that purchased their shares in a given year. For example, it might calculate that for a merger/demerger occurring on 1 July 2000, 12% of shareholders purchased their shares during 1997-98. We can then use this 12% to weight the capital gain accrued by these shareholders and then aggregate the result. The function is =ProbDist(*DistType, X1, X2, Mean, SD*) and calculates the probability under the selected distribution, in the range [X1,X2] given a mean and standard deviation.

Four distributional assumptions of the pattern of purchases for individual and institutional shareholders are available: Exponential, Gamma, Chi-square and (truncated) Normal. The user can run the model with any of the four distributions to examine the sensitivity of results to the nature of the distribution. The user can specify a mean and standard deviation of the profile of shareholdings (standard deviation only applies in Normal and Gamma distributions).

The arguments are defined as follows:

| **Argument** | **Description** | **Example** |
|---|---|---|
| *DistType* | 1 for exponential, 2 for gamma, 3 for chi-square and 4 for normal. | 1 |
| *X1* | Lower bound of the interval (a number of years). | 5 |
| *X2* | Upper bound of the interval (a number of years).<br>*Note: X2 should always be larger than X1* | 6 |
| *Mean* | Mean of the distribution (in years). | 4 |
| *SD* | Standard deviation of the distribution (gamma and normal only). | 3 |

The ProbDist(...) function returns the probability of a random variable of the selected distribution falling in the specified interval, given the mean and standard deviation. For example, =ProbDist(4,0,2,3,5) will calculate the area under a normal probability distribution in the interval [0,2] with a mean of 3 and standard deviation of 5.

The code to generate this probability is reproduced below. To see the code while the model is open in Excel 97, hit Atl+F11 and look at the code in Module1 in the model. Note that in Visual Basic a comment is preceded by the character ' (single quote), and an _ (underscore) is used to wrap long lines.

```
Function ProbDist(DistType, X1, X2, Mean, SD)
                'the function to calculate probabilities in a given interval
  Application.Volatile (False)
                'helps control automatic calculation
  If DistType = 1 Then
    ProbDist = Exp(-X1 / Mean) - Exp(-X2 / Mean)
                'exponential assumption (function gives right hand tail, hence X1-X2)
  ElseIf DistType = 2 Then
    ProbDist = Application.WorksheetFunction.GammaDist(X2, Mean^2 / SD^2, SD^2 / Mean, True) - _
               Application.WorksheetFunction.GammaDist(X1, Mean^2 / SD^2, SD^2 / Mean, True)
                'gamma assupmtion (function gives left hand tail, hence X2-X1)
  ElseIf DistType = 3 Then
    ProbDist = Application.WorksheetFunction.ChiDist(X1, Mean) - _
               Application.WorksheetFunction.ChiDist(X2, Mean)
                'chi square assupmtion (function gives right hand tail, hence X1-X2)
  Else
    ProbDist = Application.WorksheetFunction.NormDist(X2, Mean, SD, True) - _
               Application.WorksheetFunction.NormDist(X1, Mean, SD, True)
                'normal assumption (function gives left hand tail, hence X2-X1)
  End If
End Function
```

Some brief comments appear throughout the code. Additional comments to help those not familiar with Visual Basic code:

- The `Application.Volatile (False)` command ensures that cells containing the custom function are only recalculated when cells directly referenced by the function's arguments change. This makes spreadsheet recalculation more efficient.
- Where a function like `Application.WorksheetFunction.GammaDist` appears it means we are using built-in Excel functions to help evaluate some probability distributions like gamma, chi-square and normal.
- For the gamma distribution we have to convert the mean and standard deviation into the parameters $\alpha$ and $\beta$ that are required for Excel's built-in gamma distribution function, hence the expressions `Mean^2 / SD^2` and `SD^2 / Mean` respectively for the gamma distribution.

## The Rescale(...) function

The Rescale(...) function corrects for the problem that all four of the probability distributions available in the ProbDist(...) function have infinite right-hand tails and the normal distribution also has an infinite left-hand tail. Hence, there will often be some non-zero probability beyond the relevant range we are interested in. The Rescale(...) function calculates the amount of the probability distribution that actually falls within the relevant range for rescaling purposes. In practice, the relevant range is usually from the current year back to 1985-86 (the model separately treats shares purchased in 1984-85 or earlier).

For example, suppose we are using an exponential distribution for shareholdings purchased in the 15 years from 1999-00 back to the first post-CGT year of 1985-86, inclusive, with a mean of 5 years. The area under this curve only adds up to 95.02% (because we are truncating the right-hand tail at 1985-86), rather than the 100% we usually require of a probability distribution, hence we scale up the post-CGT weights by dividing by 0.9502.

The Excel custom function that calculates this rescaling factor is =Rescale(*DistType, X, Mean, SD*) which calculates the probability in the relevant range for rescaling purposes. The arguments are as follows:

| Argument | Description | Example |
|---|---|---|
| *DistType* | 1 for exponential, 2 for gamma, 3 for chi-square and 4 for normal. | 1 |
| *X* | Upper bound of the interval. | 15 |
| *Mean* | Mean of the distribution. | 4 |
| *SD* | Standard deviation of the distribution (gamma and normal only). | 3 |

The Rescale(...) function returns the probability of a random variable of the selected distribution falling between 0 and the specified upper bound, given the mean and standard deviation. For example, =Rescale(2,15,4,5) will calculate the area under a gamma probability distribution in the interval [0,15] with a mean of 4 and standard deviation of 5. The code to generate this probability is reproduced below. The additional comments for the ProbDist(...) function above also apply here.

```
Function Rescale(DistType, X, Mean, SD)
                'the function to calculate cumulative probabilities within range
  Application.Volatile (False)
                'helps control automatic calculation
  If DistType = 1 Then
    Rescale = 1 - Exp(-X / Mean)
                'exponential assumption
  ElseIf DistType = 2 Then
    Rescale = Application.WorksheetFunction.GammaDist(X, Mean^2 / SD^2, SD^2 / Mean, True)
                'gamma assupmtion
  ElseIf DistType = 3 Then
    Rescale = 1 - Application.WorksheetFunction.ChiDist(X, Mean)
                'chi square assupmtion
  Else
    Rescale = Application.WorksheetFunction.NormDist(X, Mean, SD, True) - _
              Application.WorksheetFunction.NormDist(0, Mean, SD, True)
                'normal assumption - has to take account of infinite left-hand tail
  End If
End Function
```

## The TheMaxOf(...) Function

This is a trivial function used to calculate a maximum from two alternatives. Excel also has a built-in MAX(...) function but this one is easier to access from within Visual Basic:

```
Function TheMaxOf(X, Y)
  If X > Y Then
    TheMaxOf = X
  Else
    TheMaxOf = Y
  End If
End Function
```

## The CG_Disp(...) function

The CG_Disp(...) function calculates the capital gains realisation ratio when a share-swap merger or demerger occurs and is treated as a disposal for CGT purposes (that is, the current treatment). This function forms the first part of calculating the capital gains realised under the current tax policy – the 'baseline case'.

Calculating the capital gains realised after a share-swap merger/demerger, considered as disposal of shares (that is, under the current policy), as a ratio of the post-merger/post-demerger value of the company requires the following:

- For each year from 1985-86 to the current year repeat the following steps (note that shares purchased prior to 1985-86 are, of course, excluded from these calculations):
  - Estimate the proportion of shareholdings that were purchased during the year in question.
  - Derive the original purchase price of the shares during the year in question.
  - Inflation-adjust the original purchase price, if they were purchased more than one year before disposal.
  - Calculate the percentage capital gain for the year in question by divided the post-merger/demerger price by the inflation-adjusted purchase price.
  - Multiply the percentage capital gain by the proportion of shares purchased during the year in question.

In effect, we have to calculate a weighted sum of the capital gains realised in each point in time, weighted by the proportion of shares originally purchased in each previous point in time.

Given a distribution of when shareholders originally purchased shares, historical share prices, CPI data and an assumption about merger/demerger gains, it is possible can calculate the capital gain realised ratio due to the merger/demerger. There are several real-life considerations to include in this:

- All share-swap mergers and demergers (that is, disposals of shares caused by a merger or demerger) occur on 1 July each year.
- All purchases and disposals occurring through normal trading of shares are made on 31 December of each year (note that this is the mid-point of each financial year, which is the most appropriate point for calculating gains between financial years).
- The share price in the June quarter immediately preceding the merger/demerger, plus any merger/demerger gain arising due to the merger/demerger, are used to calculate the share price at the time of the merger/demerger. (This is the disposal price for calculating the realised gains under current CGT arrangements.)
- The share price is normalised to $1 on the day of the merger/demerger (the model uses the percentage change rather than absolute change, so this is a fairly trivial assumption).
- The June quarter average of the All Ordinaries Index is used as a proxy for a generic share price up until June quarter 1998, then a capital growth assumption is applied thereafter. Users can readily enter alternative historical data into the model to allow for the possibility that takeover targets may under-perform the market average.
- The June quarter 8 capital cities CPI is used for the inflation adjustment up until June quarter 1998; an inflation assumption is applied thereafter.
- Pre-1985 shareholdings are excluded from the calculation.
- The further away we get from 1985, the smaller the proportion of pre-1985 shareholdings.
- Standard continuous probability distributions (such as exponential, gamma, chi-square and normal) all tend to be virtually zero once you get about two standard deviations from the mean. To fatten up the tails we mix in a uniform distribution. This gives more flexible distributions that can produce profiles such as a large amount of turnover within a few recent years, but also a reasonable proportion of long-term shareholders.
  *Note: there are charts included in the model that illustrate the probability distribution/shareholding profile.*
- Shares purchased within 12 months of the merger/demerger do not qualify for CPI indexation, while those purchased earlier require CPI indexation.
- The acquisition price used for shares purchased on 31 December is the average of the All Ordinaries in the two June quarters either side of the December acquisition. This is scaled relative to the $1 disposal price. Note: This and the previous step are used to work out ratios of the disposal price to the acquisition price. Both are subsequently scaled up to the actual dollar disposal values using the assumed level of share-swap merger and demerger activity.
- The CPI adjustment uses the June quarter CPI immediately preceding the 1 July merger, relative to the average of the CPI in the two June quarters either side of the December acquisition (unless the acquisition was within 12 months).

The Excel custom function that calculates the capital gain when a share-swap merger or demerger is treated as a disposal is:

=CG_Disp(*SharePrice, Inflation, CurrentYr, StartYr, CgtStartYr, Dist, Mean, StDev, UniProp, Pre85Prop, Pre85Decl, AbnormalGain*)

There are several arguments to this function, with the following usage:

| Argument | Description | Example |
|---|---|---|
| *SharePrice* | The range of cells containing the June qtr All Ords timeseries | $D$10:$AP$10 |
| *Inflation* | The range of cells containing the June qtr CPI timeseries<br>Note: the *SharePrice* range and *Inflation* range should be the same dimension. | $D$12:$AP$12 |
| *CurrentYr* | The year in which share-swap mergers/demergers force a disposal | 25 |
| *StartYr* | The first year in which mergers/demergers can occur (2000-01) | 21 |
| *CgtStartYr* | The year where CG first started to accrue (1985-86)<br>Note: the above three ... *Yr* arguments index the relevant cell in the *SharePrice* range. | 6 |
| *Dist* | The probability distribution to use - see ProbDist(...) function above | 1 |
| *Mean* | The mean of the probability distribution | 5 |
| *StDev* | The standard deviation (gamma & normal only) | 4 |
| *UniProp* | The proportion of uniform distribution to mix in to get fatter tails | 0.2 |
| *Pre85Prop* | The proportion of shares purchased pre-1985 in the *StartYr* | 0.1 |
| *Pre85Decl* | The rate of decline in *Pre85Prop* as we move away from *StartYr* | 0.01 |
| *AbnormalGain* | The percentage merger/demerger gain | 0.15 |

The CG_Disp(...) function returns the capital gain realised, measured as a proportion of the total value of the merger/demerger activity. The actual function is reproduced below, followed by additional comments to help interpret the Visual Basic code. Note that the ProbDist(...), Rescale(...) and TheMaxOf(...) functions are all used within the CG_Disp(...) function.

```
Function CG_Disp(SharePrice, Inflation, CurrentYr, StartYr, CgtStartYr, Dist, Mean, StDev, _
                 UniProp, Pre85Prop, Pre85Decl, AbnormalGain)
                 'the function to calculate CG realised for a share-swap merger/demerger
                 'treated as a disposal
  Dim i, ScaleFactor, Weight, PurchasePrice, CpiAdjPurPrice, ExcludeProp, _
      MergerSharePrice, MergerCPIlevel, AdjStartYr
                 'defines some temporary variables
  Application.Volatile (False)
                 'helps control automatic calculation
  AdjStartYr = TheMaxOf(CgtStartYr, CurrentYr - 20)
                 'adjust start year so 20 years is maximum length of share holding
  ScaleFactor = Rescale(Dist, CurrentYr - AdjStartYr, Mean, StDev)
                 'calculates the density between disposal year and 1985
  ExcludeProp = (1 - UniProp - TheMaxOf(0, Pre85Prop - Pre85Decl * (CurrentYr - StartYr)))
                 'proportion of Pre85 and uniform distribution to subtract from the distribution
  MergerSharePrice = SharePrice(CurrentYr - 1) * (1 + AbnormalGain)
  MergerCPIlevel = Inflation(CurrentYr - 1)
                 'calculates the share price and the CPI for a 1 July merger using June qtr data
  For i = AdjStartYr To CurrentYr - 1
                 'loop year by year from 1985-86 to the year before the 1 July disposal
    Weight = ProbDist(Dist, CurrentYr - i - 1, CurrentYr - i, Mean, StDev) / ScaleFactor * _
                      ExcludeProp + UniProp / (CurrentYr - AdjStartYr)
                 'the proportion of people who bought their shares in a given year
    PurchasePrice = 0.5 * (SharePrice(i - 1) + SharePrice(i)) / MergerSharePrice
                 'backcasting a share price for the year in which the shares were purchased
    If i = CurrentYr - 1 Then
                 'check for length of share holdings
      CpiAdjPurPrice = PurchasePrice
                 'no inflation adjustment for holding periods < 1 year
    Else
      CpiAdjPurPrice = PurchasePrice * MergerCPIlevel / (0.5 * (Inflation(i - 1) + Inflation(i)))
                 'adjusting the purchase price for inflation if >= 1 year
    End If
    CG_Disp = CG_Disp + Weight * (1 - CpiAdjPurPrice)
                 'weighted accumulation of capital gain, note that disposal price = 1
  Next i
End Function
```

Notes:

- The command `Dim i, ScaleFactor, ...` creates temporary variables to store intermediate calculations.
- The proportion of pre-85 shares, `Pre85Prop` is reduced by the amount `Pre85Decl` each year until it hits 0, so that the further we move away from 1985, the smaller the pre-85 proportion.

The calculations made by the above custom function are represented in a (perhaps more familiar) spreadsheet presentation in the graphic below. It is an illustration only (not actually part of the model) to help readers not

comfortable with Visual Basic code. Variable names in the illustration below typed in `Courier` font correspond to variables in the custom function code above. Variable names typed in *Time New Roman Italics* font are extra variables included to make the illustration more transparent. If you would like to see the formulae behind each cell, see the file **Example.xls** accompanying the electronic copy of the model.

This graphic calculates the capital gains realisation ratio (as a proportion of the total market value of the merger) under current tax policies, using an exponential distribution with a mean of 5. The merger is assumed to occur on 1 July 2000. Capital gain liable shares in the target company are purchased on 31 December each year, starting with 31 December 1985 with the last purchase occurring on 31 December 1999.

The result of the calculation is 28.054%. This means that a company with a market value of $10m after the merger/demerger (that is inclusive of abnormal gains) would generate $2.8054m in realised capital gains in the year the merger/demerger occurred.

Note that the calculation in the graphic below takes up a large number of cells to calculate the capital gains realised for a single year. To estimate an NPV we need to repeat this calculation for many years. This is where custom functions can be far more efficient – taking up only 1 cell per year of calculation, rather than the scores of cells that would be used if it was done using standard spreadsheet techniques. Most of the cells in the graphic can be removed and simply replaced with one cell containing:

=CG_Disp($C$7:$C$22, $D$7:$D$22, 17, 17, 2, 1, 5, 5, 0.2, 0.1, 0.01, 0.15)

CG_Disp | =SUM(L8:L22)

| | A | B | C | D | E | F | G | H | I | J | K | L |
|---|---|---|---|---|---|---|---|---|---|---|---|---|
| 1 | | AbnormalGain: | 15.0% | | | ScaleFactor: | 95.02% | | | | | |
| 2 | | UniProp: | 20.0% | | | MergerSharePrice: | 5.103 | | | | | |
| 3 | | Pre85Prop: | 10.0% | | | MergerCPIlevel: | 2.488 | | | | | |
| 4 | | ExcludeProp: | 70.0% | | | CG_Disp: | 28.054% | | | | | |
| 5 | | | | | | | | | | | | |
| 6 | *Year* | *Year Index* | Share Price | Inflation | Prob Dist | Weight | *Nominal Purchase Price* | Purchase Price | *Inflation at time of purchase* | CpiAdj PurPrice | *Capital gain* | *Weighted capital gain* |
| 7 | 1984-85 | 1 | 1.218 | 1.364 | na | na | na | na | na | na | na | na |
| 8 | 1985-86 | 2 | 1.694 | 1.479 | 1.102% | 2.145% | 1.456 | 0.285 | 1.422 | 0.499 | 0.501 | 0.011 |
| 9 | 1986-87 | 3 | 2.510 | 1.616 | 1.346% | 2.325% | 2.102 | 0.412 | 1.548 | 0.662 | 0.338 | 0.008 |
| 10 | 1987-88 | 4 | 2.102 | 1.732 | 1.644% | 2.545% | 2.306 | 0.452 | 1.674 | 0.671 | 0.329 | 0.008 |
| 11 | 1988-89 | 5 | 2.113 | 1.863 | 2.009% | 2.813% | 2.108 | 0.413 | 1.797 | 0.572 | 0.428 | 0.012 |
| 12 | 1989-90 | 6 | 2.103 | 2.006 | 2.453% | 3.141% | 2.108 | 0.413 | 1.934 | 0.531 | 0.469 | 0.015 |
| 13 | 1990-91 | 7 | 2.120 | 2.074 | 2.996% | 3.541% | 2.112 | 0.414 | 2.040 | 0.505 | 0.495 | 0.018 |
| 14 | 1991-92 | 8 | 2.305 | 2.100 | 3.660% | 4.029% | 2.212 | 0.434 | 2.087 | 0.517 | 0.483 | 0.019 |
| 15 | 1992-93 | 9 | 2.396 | 2.139 | 4.470% | 4.626% | 2.351 | 0.461 | 2.119 | 0.541 | 0.459 | 0.021 |
| 16 | 1993-94 | 10 | 2.901 | 2.176 | 5.460% | 5.355% | 2.648 | 0.519 | 2.158 | 0.598 | 0.402 | 0.022 |
| 17 | 1994-95 | 11 | 2.833 | 2.274 | 6.669% | 6.246% | 2.867 | 0.562 | 2.225 | 0.628 | 0.372 | 0.023 |
| 18 | 1995-96 | 12 | 3.175 | 2.344 | 8.145% | 7.334% | 3.004 | 0.589 | 2.309 | 0.634 | 0.366 | 0.027 |
| 19 | 1996-97 | 13 | 3.572 | 2.352 | 9.948% | 8.662% | 3.373 | 0.661 | 2.348 | 0.700 | 0.300 | 0.026 |
| 20 | 1997-98 | 14 | 3.840 | 2.368 | 12.151% | 10.285% | 3.706 | 0.726 | 2.360 | 0.765 | 0.235 | 0.024 |
| 21 | 1998-99 | 15 | 4.128 | 2.427 | 14.841% | 12.266% | 3.984 | 0.781 | 2.398 | 0.810 | 0.190 | 0.023 |
| 22 | 1999-00 | 16 | 4.437 | 2.488 | 18.127% | 14.687% | 4.283 | 0.839 | na | 0.839 | 0.161 | 0.024 |
| 23 | | | | | | | | | | | | |

CG_Disp | CG_Acqu | Ready | NUM

## The CG_Acqu(...) function

The CG_Acqu(...) function calculates the capital gains realisation ratio when shares acquired during share-swap mergers or demergers are subsequently disposed of voluntarily. This function forms the second part of calculating the capital gains realised under the current tax policy – the 'baseline case'. This calculation is necessary to ensure the modelling fully captures the gross flows under the alternative policies.

For example, calculating the capital gains realisation ratio from normal disposals of shares occurring in the year 2004-05: The scope only extends to disposals of shareholdings that were originally acquired through a share-swap merger/demerger that took place on or since the policy simulation start date of 1 July 2000. Shareholders dispose of their shares over time according to some probability distribution. To calculate the capital gains realisation ratio take, for example, the proportion of shares acquired on 1 July 2000 and disposed on 31 December 2004 multiplied by the capital gain accrued over that period. Added to this is the proportion of shares acquired on 1 July 2000 and disposed on 31 December 2004 times the gains over that period, and so on, finishing with a merger on 1 July 2004 disposed of on 31 December 2004. There are several real-life factors we need to add to this:

- There are no pre-1985 shareholdings to account for, as under the current treatment all shares are considered to be acquired on the day of the merger, which is 1 July 2000 or later. Under current policies, all pre-85 status is lost after a merger or demerger.
- The probability distributions are the same as those in the CG_Disp(...) function (exponential, gamma, chi-square and normal) with a uniform distribution mixed in to fatten up the tails.
- All shares are assumed to have been disposed of by the end of 20 years (the area under the probability distribution in the range [0,20] = 100%).
- Shares purchased within 12 months of the merger/demerger do not qualify for CPI indexation.
- Timing assumption: disposals caused by mergers/demergers always occur on 1 July of each financial year. Disposals via normal share trading always occur on the midpoint, 31 December of each financial year.
- The acquisition price for shares acquired in a July 1 merger/demerger is the average level of the All Ordinaries index in the June quarter immediately preceding the merger/demerger, normalised to $1.
- The disposal price used for shares sold on 31 December is the average of the All Ordinaries in the two June quarters either side of the December disposal. This is scaled relative to the original $1 acquisition price.
- An allowance is made for higher turnover in shares in the first year after a merger/demerger due to shareholders rationalising their holdings.
- Note that merger/demerger gains do not need to be accounted for here as the capital gain arising from any abnormal gain was already counted in the CG_Disp(...) function. It is assumed that the abnormal gain counted in CG_Disp(...) function captures the additional earnings potential of the merged/demerged company. That is, the abnormal gain captures the efficiency gains from the merger/demerger.
- The CPI adjustment uses the June quarter CPI immediately preceding the 1 July merger/demerger, relative to the average of the CPI in the two June quarters either side of the December acquisition (unless the acquisition was within 12 months).

The Excel custom function that calculates the ongoing realisations of capital gains when a share-swap merger/demerger is treated as a disposal is:
=CG_Acqu(*SharePrice, Inflation, CurrentYr, StartYr, Dist, Mean, StDev, UniProp, FirstYrProp*)

There are several arguments to this function, which have the following usage:

| Argument | Description | Example |
|---|---|---|
| *SharePrice* | The range of cells containing the June qtr All Ords timeseries | $D$10:$AP$10 |
| *Inflation* | The range of cells containing the June qtr CPI timeseries<br>Note: the *SharePrice* range and *Inflation* range should be the same dimension. | $D$12:$AP$12 |
| *CurrentYr* | The year in which the normal-trading disposals occur | 25 |
| *StartYr* | The first year of mergers/demergers causing disposals (2000-01) | 21 |

| | | |
|---|---|---|
| | Note: the above two ... *Yr* arguments index the relevant cell in the *SharePrice* range. | |
| *Dist* | The probability distribution to use - see ProbDist(...) function above | 1 |
| *Mean* | The mean of the probability distribution | 5 |
| *StDev* | The standard deviation (gamma & normal only) | 4 |
| *UniProp* | The proportion of uniform distribution to mix in to get fatter tails | 0.2 |
| *FirstYrProp* | Extra turnover in shares in the first year due to rationalisation | 0.1 |

The CG_Acqu(...) function returns the capital gain realised in a given year, measured as a proportion of the total value of the company, post merger/demerger, when the share were considered disposed of then newly acquired after the merger/demerger. The actual function is reproduced below.

```
Function CG_Acqu(SharePrice, Inflation, CurrentYr, StartYr, Dist, Mean, StDev, UniProp)
                'the function to calculate ongoing CG after a share-swap merger/corporate
recontruction
                'treated as a disposal
  Dim j, ScaleFactor, Weight, DisposalPrice, CpiAdjPurPrice, AdjStartYr, _
      CurrentSharePrice, CurrentCPIlevel
                'defines some temporary variables
  Application.Volatile (False)
                'helps control automatic calculation
  AdjStartYr = TheMaxOf(StartYr, CurrentYr - 19)
                'adjust start year so 20 years is maximum length of share holding
  ScaleFactor = Rescale(Dist, 20, Mean, StDev)
                'rescales for any density prior to the start year (really for normal dist only)
  CurrentSharePrice = 0.5 * (SharePrice(CurrentYr - 1) + SharePrice(CurrentYr))
  CurrentCPIlevel = 0.5 * (Inflation(CurrentYr - 1) + Inflation(CurrentYr))
                'calculates current share price and CPI for a 31 Dec sale of shares
  For j = AdjStartYr To CurrentYr
                'loop year by year from the first year of the simulation period to the current year
    Weight = ProbDist(Dist, CurrentYr - j, CurrentYr - j + 1, Mean, StDev) / ScaleFactor * _
             (1 - UniProp - FirstYrProp) + UniProp / 20 + IIf(j = CurrentYr, FirstYrProp, 0)
                'the proportion of people who bought their shares in a given year
    DisposalPrice = CurrentSharePrice / SharePrice(j - 1)
                'forecasting a share price for the year in which the shares were disposed
    If CurrentYr = j Then
                'check for length of share holdings
      CpiAdjPurPrice = 1
                'no inflation adjustment for holding periods < 1 year
    Else
      CpiAdjPurPrice = CurrentCPIlevel / Inflation(j - 1)
                'adjusting the purchase price for inflation if >= 1 year
    End If
    CG_Acqu = CG_Acqu + Weight * (DisposalPrice - CpiAdjPurPrice)
                'weighted accumulation of capital gain
  Next j
End Function
```

The calculations made by the above custom function are represented in a (perhaps more familiar) spreadsheet presentation in the graphic below. It is an illustration only (not actually part of the model) to help readers not comfortable with Visual Basic code.

This graphic calculates the capital gains realised (as a percentage of the total market value of the merger/demerger) under current tax policies, using an exponential distribution with a mean of 5. The year of evaluation is assumed to be 2004-05. Capital gain liable shares in the companies were acquired through various mergers/demergers on 1 July each year from 2000 to 2004.

The result of the calculation is 6.935%. This mean that if the average company is valued at $10m after a merger/demerger, then all mergers/demergers that occurred from 1 July 2000 to 1 July 2004 would generate $0.6935m in realised capital gains during 2004-05.

Note again that the calculations in the graphic below takes up a large number of cells compared to the custom function:

=CG_Acqu($C$7:$C$12, $D$7:$D$12, 6, 2, 1, 5, 5, 0.2, 0)



CG_Acqu | =SUM(K8:K12)

| | A | B | C | D | E | F | G | H | I | J | K | L |
|---|---|---|---|---|---|---|---|---|---|---|---|---|
| 1 | | UniProp: | 20.0% | | CurrentSharePrice: | | 6.1483 | | | | | |
| 2 | | ScalePactor: | 98.168% | | CurrentCPIlevel: | | 2.7804 | | | | | |
| 3 | | | | | | | | | | | | |
| 4 | | | | | | CG_Acqu: | 6.935% | | | | | |
| 5 | | | | | | | | | | | | |
| 6 | | *Year* | *Year Index* | Share Price | Infla-tion | Prob Dist | Weight | Disposal Price | *Nominal Purchase Price* | CpiAdj PurPrice | *Capital gain* | *Weighted capital gain* |
| 7 | | 1999-00 | 1 | 4.437 | 2.488 | n.a. | n.a. | n.a. | n.a. | n.a. | n.a. | n.a. |
| 8 | | 2000-01 | 2 | 4.770 | 2.550 | 8.145% | 7.638% | 1.386 | 1.000 | 1.118 | 0.268 | 0.020 |
| 9 | | 2001-02 | 3 | 5.128 | 2.614 | 9.948% | 9.107% | 1.289 | 1.000 | 1.090 | 0.199 | 0.018 |
| 10 | | 2002-03 | 4 | 5.513 | 2.679 | 12.151% | 10.902% | 1.199 | 1.000 | 1.064 | 0.135 | 0.015 |
| 11 | | 2003-04 | 5 | 5.926 | 2.746 | 14.841% | 13.094% | 1.115 | 1.000 | 1.038 | 0.078 | 0.010 |
| 12 | | 2004-05 | 6 | 6.371 | 2.815 | 18.127% | 15.772% | 1.038 | 1.000 | n.a. | 0.037 | 0.006 |
| 13 | | | | | | | | | | | | |
| 14 | | | | | | | | | | | | |
| 15 | | | | | | | | | | | | |
| 16 | | | | | | | | | | | | |
| 17 | | | | | | | | | | | | |
| 18 | | | | | | | | | | | | |
| 19 | | | | | | | | | | | | |
| 20 | | | | | | | | | | | | |
| 21 | | | | | | | | | | | | |
| 22 | | | | | | | | | | | | |
| 23 | | | | | | | | | | | | |



## The CG_Roll(…) function

The CG_Roll(..) function calculates the capital gains realisation ratio when shares acquired voluntarily in the past and exchanged for shares in another company via a share-swap merger or demerger at some time on or since 1 July 2000 (with capital gains rolled-over allowed), are disposed of voluntarily in future years. This function calculates the capital gains realised under the proposed change to tax policy – the 'alternative case'.

For example, consider calculating the capital gains realisation ratio from disposals via normal trading of shares occurring in the year 2004-05 that were swapped in a share-swap merger or demerger that took place on or since the policy simulation start date of 1 July 2000. Shareholders dispose of their shares over time according to some probability distribution. To calculate the total capital gains realised, we calculate the relevant capital gain in each cross-combination of acquisition and disposal and multiply it by the proportion of shares in that cross-combination. All shares are assumed to be disposed of in the midpoint of the year, for example, for 2004-05 on 31 December 2004:

- The first batch to consider is shares swapped in a merger/demerger on 1 July **2000:**
  - acquired on 31 December **1985**, swapped on 1 July **2000** and disposed on 31 December 2004,
  - acquired on 31 December **1986**, swapped on 1 July **2000** and disposed on 31 December 2004,
  - … and so on, up to shares:
  - acquired on 31 December **1999**, swapped on 1 July **2000** and disposed on 31 December 2004.
- The next batch to consider is shares swapped in a merger/demerger on 1 July **2001:**
  - acquired on 31 December **1985**, swapped on 1 July **2001** and disposed on 31 December 2004,
  - … and so on, up to shares:
  - acquired on 31 December **2000**, swapped on 1 July **2001** and disposed on 31 December 2004.
- And so on until we reach the final batch to consider, shares swapped on 1 July **2004:**
  - acquired on 31 December **1985**, swapped on 1 July **2004** and disposed on 31 December 2004,
    and so on up to shares:
  - acquired on 31 December **2003**, swapped on 1 July **2004** and disposed on 31 December 2004.

In effect, this is like the cross product of the CG_Disp(...) function and the CG_Acqu(...) functions discussed on pages 7 and 11, respectively.

The Excel custom function that calculates the capital gain when a share-swap merger is treated as a disposal:
=CG_Roll(*SharePrice, Inflation, CurrentYr, StartYr, CgtStartYr, _*
*PreDist, PreMean, PreStDev, PreUniProp, Pre85Prop, Pre85Decl, GrandFather, _*
*PostDist, PostMean, PostStDev, PostUniProp, FirstYrProp, AbnormalGain*)

There are several arguments to this function, which have the following usage:

| Argument | Description | Example |
|---|---|---|
| *SharePrice* | The range of cells containing the June qtr All Ords timeseries | $D$10:$AP$10 |
| *Inflation* | The range of cells containing the June qtr CPI timeseries<br>Note: the *SharePrice* range and *Inflation* range should be the same dimension. | $D$12:$AP$12 |
| *CurrentYr* | The year in which the disposals via normal trading occur | 25 |
| *StartYr* | The first year in which share-swap mergers can occur (2000-01) | 21 |
| *CgtStartYr* | The year where CG first started to accrue (1985-86)<br>Note: the above three ... *Yr* arguments index the relevant cell in the *SharePrice* range. | 6 |
| *PreDist* | Pre-merger prob. dist. to use - see ProbDist(...) function above | 1 |
| *PreMean* | Pre-merger mean of the probability distribution | 5 |
| *PreStDev* | Pre-merger standard deviation (gamma & normal only) | 4 |
| *PreUniProp* | Pre-merger prop. of uniform dist. to mix in to get fatter tails | 0.2 |
| *Pre85Prop* | The proportion of shares purchased pre-1985 in the *StartYr* | 0.1 |
| *Pre85Decl* | The rate of decline in *Pre85Prop* as we move away from *StartYr* | 0.01 |
| *GrandFather* | TRUE if pre-85 share keep their status, FALSE is they lose pre-85 | FALSE |
| *PostDist* | Post-merger prob. dist. to use - see ProbDist(...) function above | 1 |
| *PostMean* | Post-merger mean of the probability distribution | 5 |
| *PostStDev* | Post-merger standard deviation (gamma & normal only) | 4 |
| *PostUniProp* | Post-merger prop. of uniform dist. to mix in to get fatter tails | 0.2 |
| *FirstYrProp* | Extra turnover in shares in the first year due to rationalisation | 0 |
| *AbnormalGain* | The percentage merger/demerger gain due to a successful merger | 0.15 |

The CG_Roll(...) function returns the capital gain realised in a given year, measured as a proportion of the total post merger market capitalisation, when the capital gains are rolled over at the time of the merger. The actual function is reproduced below.

```
Function CG_Roll(SharePrice, Inflation, CurrentYr, StartYr, CgtStartYr, _
                 PreDist, PreMean, PreStDev, PreUniProp, Pre85Prop, Pre85Decl, GrandFather, _
                 PostDist, PostMean, PostStDev, PostUniProp, FirstYrProp, AbnormalGain)
                'the function to calculate CG realised when a share swap/demerger is rolled over
  Dim i, j, ScaleFactori, ScaleFactorj, Weighti, Weightj, PurchasePrice, DisposalPrice, _
      CpiAdjPurPrice, ExcludeProp, AdjStartYrj, AdjStartYri, CurrentSharePrice, CurrentCPIlevel
                'defines some temporary variables
  Application.Volatile (False)
                'helps control automatic calculation
  AdjStartYrj = TheMaxOf(StartYr, CurrentYr - 19)
                'adjust start years so 20 years is maximum length of shareholding
  CurrentSharePrice = 0.5 * (SharePrice(CurrentYr - 1) + SharePrice(CurrentYr))
  CurrentCPIlevel = 0.5 * (Inflation(CurrentYr - 1) + Inflation(CurrentYr))
                'calculates current share price and CPI for a 31 Dec sale of shares
  ScaleFactorj = Rescale(PostDist, 20, PostMean, PostStDev)
                'rescales for any density prior to the start year or after year 20
  For j = AdjStartYrj To CurrentYr
                'loop year by year from the first year of the simulation period to the current year
    AdjStartYri = TheMaxOf(CgtStartYr, j - 20)
                'adjust start years so 20 years is maximum length of share holding
    ScaleFactori = Rescale(PreDist, j - AdjStartYri, PreMean, PreStDev)
                'calculates the density between disposal year and 1985
    Weightj = ProbDist(PostDist, CurrentYr - j, CurrentYr - j + 1, PostMean, PostStDev) / _
              ScaleFactorj * (1 - PostUniProp - FirstYrProp) + PostUniProp / 20 + _
              IIf(j = CurrentYr, FirstYrProp, 0)
                'the proportion of people who bought their shares in a given year
    ExcludeProp = 1 - PreUniProp - TheMaxOf(0, Pre85Prop - Pre85Decl * (j - StartYr))
                'the proportion of Pre85 shareholding and uniform distribution to subtract
    DisposalPrice = CurrentSharePrice / SharePrice(j - 1)
                'forecasting a share price for the year in which the shares were disposed
    For i = AdjStartYri To j - 1
                'loop year by year from 1985-86 to the year before the 1 July disposal
      Weighti = ProbDist(PreDist, j - i - 1, j - i, PreMean, PreStDev) / ScaleFactori * _
                ExcludeProp + PreUniProp / (j - AdjStartYri)
                'the proportion of people who bought their shares in a given year
      PurchasePrice = 0.5 * (SharePrice(i - 1) + SharePrice(i)) / SharePrice(j - 1) / _
                (1 + AbnormalGain)
                'backcasting a share price for the year in which the shares were purchased
      CpiAdjPurPrice = PurchasePrice * CurrentCPIlevel / (0.5 * (Inflation(i - 1) + Inflation(i)))
                'inflation adj. purchase price - all holdings are >= 1 year (30 Dec assumption)
      CG_Roll = CG_Roll + Weighti * Weightj * (DisposalPrice - CpiAdjPurPrice)
                'weighted accumulation of capital gain
    Next i
  Next j
  If GrandFather = False Then
                'allowing for the option that pre-85 status is lost after rollover
    For j = AdjStartYrj To CurrentYr
                'loop year by year from the first year of the simulation period to the current year
      Weightj = ProbDist(PostDist, CurrentYr - j, CurrentYr - j + 1, PostMean, PostStDev) / _
                ScaleFactorj * (1 - PostUniProp - FirstYrProp) + PostUniProp / 20 + _
                IIf(j = CurrentYr, FirstYrProp, 0)
                'the proportion of people who bought their shares in a given year
      DisposalPrice = CurrentSharePrice / SharePrice(j - 1)
                'forecasting a share price for the year in which the shares were disposed
      Weighti = TheMaxOf(0, Pre85Prop - Pre85Decl * (j - StartYr))
                'the proportion of people who bought their shares pre 1985
      If CurrentYr = j Then
                'check for length of share holdings
        CpiAdjPurPrice = 1
                'no inflation adjustment for holding periods < 1 year
      Else
        CpiAdjPurPrice = CurrentCPIlevel / Inflation(j - 1)
                'adjusting the purchase price for inflation if >= 1 year
      End If
      CG_Roll = CG_Roll + Weighti * Weightj * (DisposalPrice - CpiAdjPurPrice)
                'weighted accumulation of capital gain
    Next j
  End If
End Function
```

Notes:

- The CG_Disp(...) function loops over i= CG start year to the year before the current year (that is, it loops over all the years where purchases of shares could have occurred prior to a disposal occurring due to a merger/demerger). The CG_Acqu(...) function loops over j= start year to the current year (that is, all the years where share-swap mergers or demergers could have occurred leading to shares sold voluntarily in the current year). The CG_Roll(...) function loops over both i and j because there is no merger/demerger causing disposals nor acquisitions to all occur at a fixed point in time – that is, it is necessary to loop over every possible combination of years when purchases and sales via normal trading of shares could have occurred.
- We use the product of the weights `Weighti * Weightj` from the pre-merger purchase profile and the post-merger disposal profile, respectively. This implicitly assumes the two distributions are independent.
- If shares lose their pre-85 status, then the capital gains are calculated only from the time of the merger/ demerger.

For the first two functions, CG_Disp(...) and CG_Acqu(...) it was possible to illustrate the calculations performed by the custom function in a spreadsheet presentation. However, the CG_Roll(...) function is roughly the cross-product of the first two and takes up so many cells that a spreadsheet illustration is just not possible.

## *Part 2 of the model – tax revenue*

The calculations for Part 2 are more straightforward, compared with the capital gains realisation ratio custom functions in the previous section. These calculations are implemented using more standard spreadsheet calculations. The types of issues taken into account in Part 2 include:

- Calculating the capital gains realisation ratios using the functions developed in Part 1.
- Grossing up realisation ratios by the amount of market capitalisation of target/reconstructed companies covered including adjustments for any cash component in transactions.
- Allowing for an increased number of share-swap mergers or demergers following the extension of rollover relief. Also allowing for an increase in share-swap mergers that would otherwise have been cash mergers.
- Allowing for a reduction in the average merger/demerger gain after rollover relief is allowed.
- Allowing for increase average length of shareholding due to the lock-in effect of rollover relief.
- Allowing for private/institutional mix of shareholders, including:
  - Different disposal/acquisition profiles for private versus institutional investors.
- Allowing for revenue collected through the capital gains tax net and the income tax net, including:
  - Profits from trading in shares that are treated on the revenue account (rather than the capital account).
  - Any tax obtained from foreign investors.
  - Increased dividends as implied by the merger/demerger gain.
- Apply average tax rates, allowing for the different average tax rates for private shareholders, superfunds, life funds, retail finds, etc...
- Calculating the net revenue impact in each year and the Net Present Value of the revenue impact.

There are two sets of calculations in the models – the share-swap merger calculations are on the worksheet titled MerCalc and the demerger calculations are on the worksheet titled DemCalc. The assumptions underpinning the models are on the worksheet titled Assump. Readers are encouraged to have the model up on their computer screen while reading the discussion below.

## Capital gains realisation ratio

To apply the custom formulae developed in Part 1 we need a *SharePrice* time series and an *Inflation* time series. Historical data for these are entered on rows 6 and 7 of the Index worksheet, respectively. The All Ordinaries share price index and the CPI are used in history, but users can feed in alternative data for share prices if they wish. On row 10 and 12 these series are rebased and the capital growth and inflation assumptions from the Assump worksheet are applied to extrapolate these series into the future.

To evaluate the current policy, we use the CG_Disp(...) formula in rows 8 and 9 for private investors and institutions, respectively, to get the capital gains realisation ratio due to share-swap mergers (on the MerCalc worksheet) and similarly for demergers (on the DemCalc worksheet) as a proportion of market capitalisation. On-going CGT realisations, post merger/demerger, are calculated using the CG_Acqu(...) formula in rows 11 and 12, again separately for private and institutions respectively. To evaluate revenue under current policies, it is also necessary to count capital gains on companies that would not have merged/reconstructed under current policies, but would have under rollover relief, to properly account for the gross flows. To do this we can use the CG_Roll(..) formula, with a merger/demerger gain and first-year disposal effect of zero (since no merger/demerger took place, there is no merger/demerger gain or first-year rationalisation, but otherwise the CGT calculations are the same). This is done in rows 14 and 15, for private and institutional investors respectively.

To evaluate the alternative policy, the CG_Roll(...) function is used, but this time allowing for merger/demerger gains. This is done in rows 19 and 20, for private and institutional investors respectively. An increase in the average length of share holding post merger/demerger is assumed to account for the additional lock in effect for investors granted rollover relief.

The reason for evaluating equations separately for private and institutional investors is so the model can allow for the higher share turnover profiles (lower mean holding times) that are likely for institutions compared to private investors.

## Market capitalisation

The market capitalisation is the next item calculated, which is used to gross up the capital gain realisation ratios. This is done in cells D22:G23 on the MerCalc and DemCalc worksheets. Under current policies we need to know the market capitalisation of all share-swap mergers/demergers, plus the capitalisation of mergers that would have been cash-based under current policies but would have a share-swap component if rollover relief was allowed. The calculation also has to factor in any merger/demerger gains applicable. These calculations are the number of mergers times the average market capitalisation, less the amount of toehold investment (for share-swap mergers only), as specified on the Assump worksheet.

To calculate capital gains revenue the model first calculates the total base of capital gains (rows 25 to 30 on the MerCalc and DemCalc worksheets), by taking the product of:

- The capital gain realisation ratios.
- The $m market capitalisation figures, noting that:
  - The toehold investment is excluded in the case of share-swap mergers;
  - Merger/demerger gains are included for those mergers/demergers that go ahead (note that under existing policies there will be some mergers/demergers that do not proceed due to the absence of rollover relief, hence these have no merger/demerger gain);
  - It is possible to have a cash component in a share-swap merger. Under the current policies, cash and share-swaps are both treated as disposals, so there is no difference. However, in the case of rollover relief the proportion that is taken as cash is treated as a disposal, while the remainder is treated as a rollover. The proportion is expressed as a proportion of market capitalisation including merger/demerger gains; and
  - The number of mergers/demerger occurring under rollover relief is assumed to be greater than or equal to the number occurring under current policies. When there are additional mergers/demergers

under the rollover relief scenario, the current-policy calculations still have to count the (lower) capital gains that would have been realised on these un-merged/un-demerger companies, to take account of gross flows.

- If rollover relief is granted, the average merger/demerger gain would fall – that is, acquiring companies could offer lower premiums to target company shareholders because the CGT barrier is removed.

Note that the calculation of the capital gains base on rows 25 to 30 on the MerCalc and DemCalc worksheets are intermediate calculations only and cannot be interpreted as gross capital gains.

## Revenue from capital gains

From the capital gains base calculations, the model takes account of the mix of private investors and the various types of institutional investors and the average tax rate paid by each. Note that the capital gains on shares held by share trading companies is treated as income, so are calculated separately.

Offsetting capital losses are applied at this point to reduce the average tax paid on a given dollar of capital gain. A one year lag is introduced to reflect the timing difference between tax liability incurred and tax collected.

These calculations are on rows 35 to 40 on the MerCalc and DemCalc worksheets. This is the total capital gains tax revenue collected under the alternative policies. However, there are still several income tax calculations to make.

Some revenue from the capital gain in the shares is actually collected through the inome tax net. This is the tax payable on profits from shares held by trading companies and is calculated in rows 45 and 48.

## Revenue from the efficiency dividend

The income tax revenue generated from the increased company income and hence dividend payments implied by the merger/demerger gain due to the merger or demerger is calculated in rows 53 to 60.

The issue is that whenever a merger/demerger gain occurs, it reflects the markets' expectation of higher future earnings from holding shares in the post-merger or post-demerger entity. If shares increase in value by (say) 30% as a result of a merger or demerger, standard asset pricing theory says that investors value the net present value of the expected additional after-tax flow of dividends at 30% of the current share price.

Given assumptions about merger/demerger gain (say, 30%), an average tax rate (say, 25%) and a rate of return for discounting purposes (say, 10%) we can reverse-engineer the implied dividend flow that investors are pricing into the share.

Note that the sum of the discount factors is a converging series: $\sum_{i=0}^{\infty} \frac{1}{(1+r)^i} = \frac{1+r}{r}$.

$$\begin{aligned}\text{Implied dividend flow} &= (0.30 \div (1.10 \div 0.10)) \div (1 - 0.25) \\ &= (0.30 \div 11.00) \div 0.75 \\ &= 3.64\%\end{aligned}$$

Hence in the above example investors are pricing in a before-tax earnings increase of 3.64 percentage points per annum as a result of the merger.

However, it is likely that this stream of higher profits is not uniform – profits may even be lower in the short term due to restructuring. To reflect this, a profile is entered in row 19 on the Index worksheet. A value of 100% on this line indicates the profit stream is at the 'long run' level in the year indicated. A value less than 100%, and even negative, generates the short-run profile described above. It is impossible to use historical

data to guide this assumption, as the short-run change in the pattern of profitability of target companies is subject to much external noise. Hence, a fairly conservative profile has been adopted as follows:

| Efficiency gain | Year 1 | Year 2 | Year 3 | Year 4 | Year 5 + |
|---|---|---|---|---|---|
| Time pattern | -25% | +25% | +50% | +75% | +100% |

Given the above calculations, it is then a matter of applying the implied increase in the earnings rate to the dollar value of the investment, then calculating income tax revenue on the increased dividend payments.

Note importantly that the efficiency dividend compounds over time as more mergers/demergers occur. After 4-5 years there has been a large number of additional mergers, and the increased profit and hence tax revenue generated by those mergers/demergers accumulates to a significant number.

The additional dividend payments are imputed to shareholders and the marginal tax rates of the various shareholders types are applied to derive revenue.

## Presentation of results

The model reports gross revenue figures – that is, the gross revenue collected under both the current and alternative policy – on rows 64 and 65 on the MerCalc and DemCalc worksheets. These gross figures are useful for benchmarking to historical CGT collection on all share disposals to ensure the figure falls within reasonable bounds.

The net Budget impact is of most use for assessing the impacts of the policies. The NPV of the series of net Budget impacts is also reported.

To ensure that the NPV is on a comparable scale to the annual numbers in the chart illustrating the annual net Budget impacts, the NPV is divided by the sum of the discount factors to give an 'averaged NPV'. The averaged NPV gives the average annual impact on the Budget in present value terms, so may be easier to interpret than a normal NPV (the latter is the cumulative sum so the magnitude can be difficult to interpret).

The formulae used, for a cash flow $X_i$ and an interest rate $r$ are:

$$Normal\ NPV = \sum_{i=0}^{\infty} \frac{X_i}{(1+r)^i} \tag{1}$$

$$Ave.\ annual\ NPV = \sum_{i=0}^{\infty} \frac{X_i}{(1+r)^i} \Big/ \sum_{i=0}^{\infty} \frac{1}{(1+r)^i} \tag{2}$$

Given that the denominator in the above equation converges, this can be written:

$$Ave.\ annual\ NPV = \frac{1+r}{r} \sum_{i=0}^{\infty} \frac{X_i}{(1+r)^i} \tag{3}$$

However, when using a real bond rate around 3% to 4% for discounting, the sum takes more than a hundred years to converge, hence formula (2) is actually used in practice, with the summation extending to 100 years instead of ∞.

See row 66, in particular cell V66, on the MerCalc or DemCalc worksheets if you would like to examine the averaged NPV formula closely.

The model also includes several charts: realisation ratios, budget impact and probability distributions to help illustrate some of the concepts.